Die Keplersche Fassregel

K. Gerber

Bei vielen Aufgaben, z.B. bei der Lösung von Differentialgleichungen, taucht die Schwierigkeit auf, dass Integrationen nicht durchgeführt werden können. So können z.B. die folgenden Integrale nicht mit elementaren Funktionen angeben werden:

$$\int \mathbf{e}^{(-x^2)}\,dx,\ \int \frac{\mathbf{e}^x}{x}\,dx,\ \int \frac{\sin(x)}{x}\,dx,\ \int \frac{1}{\ln(x)}\,dx$$

Oft bleibt dann nur die Möglichkeit, solche Integrale mit Hilfe von Näherungswerten für bestimmte Integrale zu tabellieren.
Die sich im Schulunterricht sofort anbietende näherungsweise Bestimmung von Flächenmaßzahlen durch Unter- und Obersummen ist für diese Zwecke rechnerisch zu aufwendig. Eine schnellere Möglichkeit ist die Approximation der zu integrierenden Funktion durch eine ganzrationale Funktion, die dann einfach zu integrieren ist. Lineare oder quadratische Funktionen eignen sich dazu besonders gut.
Wir bestimmen in diesem Arbeitsblatt quadratische Näherungsfunktionen p für eine gegebene Funktion f und ermitteln damit die Näherung $\int_a^b \mathrm{p}(x)\,dx$ für das Integral $\int_a^b \mathrm{f}(x)\,dx$.

Aufgabe 1: Ermittlung des Näherungspolynoms und Integration

Um unsere Näherungsmethode zumindest qualitativ prüfen zu können, betrachten wir zunächst eine Funktion f, die auf dem angegeben Intervall selbst geschlossen integrierbar ist:

$\mathrm{f}(x) = \sin(x)$ auf dem Intervall [a,b] mit a=0 und b=2.

a) Bestimme ein quadratisches Polynom $\mathrm{p}(x)$ ($\mathrm{p}(x) = r\,x^2 + s\,x + t$) mit

$$\mathrm{p}(a) = \sin(a),\ \mathrm{p}\left(\frac{a+b}{2}\right) = \sin\left(\frac{a+b}{2}\right),\ \mathrm{p}(b) = \sin(b)\ .$$

b) Zeichne die Graphen von f und p in ein Koordinatensystem.

c) Integriere p über [0,2] und vergleiche mit dem exakten Wert

(Wir können nun unsere Ergebnisse mit $\int_0^2 \sin(x)\,dx$ vergleichen und das Näherungsverfahren beurteilen.

Natürlich ist die Integration einer Näherungsfunktion eigentlich nur für Funktionen sinnvoll, für die wir *keine* Stammfunktionen angeben können.)

− ***Lösungsvorschlag***

```
> restart;
```

a) Eingabe der Funktion f, des Näherungspolynoms und der Intervallgrenzen:

```
> f:=x->sin(x);
> p:=x->r*x^2+s*x+t;
> a:=0; b:=2;
```

$$f := \sin$$

$$a := 0$$

$$b := 2$$

Zunächst muss das zu der Steckbriefaufgabe gehörende Gleichungssystemgelöst werden.

Die Ergebnisse werden den Variablen r,s,t zugewiesen.

```
> L:=solve({f(a)=p(a),f((a+b)/2)=p((a+b)/2),f(b)=p(b)},{r,s,t})
  ;
> assign(L);
```

$$L := \{t = 0, s = -\frac{1}{2}\sin(2) + 2\sin(1), r = -\sin(1) + \frac{1}{2}\sin(2)\}$$

Ermittlung des Polynoms:

```
> 'p(x)'=p(x);
```

$$\mathrm{p}(x) = \left(-\sin(1) + \frac{1}{2}\sin(2)\right)x^2 + \left(-\frac{1}{2}\sin(2) + 2\sin(1)\right)x$$

b) Graphen zeichnen:

```
> plot({f(x),p(x)},x=-0.5..6.5,y=-1..1);
```



c) Integration des Näherungspolynoms und Vergleich:

```
> Int('p(x)',x=a..b)=evalf(int(p(x),x=a..b));
> Int(f(x),x=a..b)=evalf(int(f(x),x=a..b));
```

$$\int_0^2 \mathrm{p}(x)\,dx = 1.425060455$$

$$\int_0^2 \sin(x)\,dx = 1.416146837$$

>

<u>Aufgabe 2:</u> Herleitung der Keplerschen Fassregel

Nun suchen wir eine Regel, die uns die mühsamen Schritte der Ermittlung des Näherungspolynoms für weitere Aufgaben erspart. Dazu betrachten wir eine beliebige, stetige Funktion f auf dem Intervall [a,b]. Wir setzen: $y1 = \mathrm{f}(a), y2 = \mathrm{f}\left(\frac{a+b}{2}\right), y3 = \mathrm{f}(b)$.

a) Bestimme die Koeffizienten des Näherungspolynoms $\mathrm{p}(x)$ mit $\mathrm{p}(x) = r\,x^2 + s\,x + t$.
b) Integriere und faktorisiere das Ergebnis.

Lösungsvorschlag

a) Gegeben ist die auf dem Intervall [a,b] stetige Funktion f.
y1, y2 und y3 sind die Funktionswerte von f an der Stelle a, (a+b)/2 und b.
Zunächst bestimmen wir die Koeffizienten eines quadratischen Näherungspolynoms p(x) durch eine "Steckbriefaufgabe".

```
> restart;
> p:=x->r*x^2+s*x+t;
```

```
> L:=solve({y1=p(a),y2=p((a+b)/2),y3=p(b)},{r,s,t});
```

$$L := \left\{r = 2\,\frac{y1 + y3 - 2\,y2}{a^2 - 2\,a\,b + b^2}, s = -\frac{a\,y1 - 4\,y2\,a + 3\,a\,y3 + 3\,b\,y1 + b\,y3 - 4\,y2\,b}{(a-b)^2}, t = \frac{a^2\,y3 + y3\,a\,b - 4\,y2\,a\,b + y1\,a\,b + y1\,b^2}{(a-b)^2}\right\}$$

Wir weisen die Ergebnisse den Variablen r,s,t zu.

```
> assign(L);
```

Damit erhalten wir das Näherungspolynom.

```
> 'p(x)'=p(x);
```

$$\mathrm{p}(x) = 2\,\frac{(y1 + y3 - 2\,y2)\,x^2}{a^2 - 2\,a\,b + b^2} - \frac{(a\,y1 - 4\,y2\,a + 3\,a\,y3 + 3\,b\,y1 + b\,y3 - 4\,y2\,b)\,x}{(a-b)^2} + \frac{a^2\,y3 + y3\,a\,b - 4\,y2\,a\,b + y1\,a\,b + y1\,b^2}{(a-b)^2}$$

b) Nun kann das Näherungspolynom in den Grenzen von a bis b integriert werden.
Es folgt die Keplersche Fassregel:

```
> Int(P(x),x=a..b)=factor(int(p(x),x=a..b));
```

$$\int_a^b \mathrm{P}(x)\,dx = -\frac{1}{6}(a-b)(y1+y3+4\,y2)$$

```
> 
```

Aufgabe 3: Anwendung

Die in Aufgabe 2 erhaltene Keplersche Fassregel kann nun verwendet werden, um Näherungswerte für bestimmte Integrale zu berechnen. Führe alle dazu notwendigen Schritte aus und berechne die in der Einleitung angegeben Integrale über sinnvollen Intervallen. Die Keplersche Fassregel liefert für quadratische Funktionen f exakte Ergebnisse. Warum?

Lösungsvorschlag

```
> restart;
```

Eingabe der Funktion f:

```
> f:=x->exp(-x^2);
```

$$f := x \to \mathbf{e}^{(-x^2)}$$

Eingabe der Keplerschen Fassregel:

```
> kepler:=(a,b)->(b-a)/6*(f(a)+4*f((a+b)/2)+f(b));
```

Berechnung des Näherungswertes:

```
> evalf(kepler(1,2));
```

$$.1346319965$$

Die Näherungeslösung für die anderen Integrale verläuft analog durch Überschreiben des Funktionsterms.

Aufgabe 4: Integration ganzrationaler Funktionen 3. Grades

Betrachte die Funktion f mit $\mathrm{f}(x) = \mathrm{x}(x^2-9)$.

a) Ermittle das quadratische Näherungspolynom p für a= -1 und b=2.

b) Zeichne die Graphen von f und p in ein Koordinatensystem.

c) Untersuche die beiden von den Graphen eingeschlossenen Differenzflächen. Was stellst du fest?

d) Kann die Vermutung verallgemeinert werden?

Lösungsvorschlag

```
> restart;
```

a) Eingabe der Funktion f, des Näherungspolynoms und der Intervallgrenzen:

```
> f:=x->x*(x^2-9);
> p:=x->r*x^2+s*x+t;
> a:=-1;b:=2;
```

$$f := x \to x\,(x^2-9)$$

$$p := x \to r\,x^2 + s\,x + t$$

$$a := -1$$

$$b := 2$$

Hier muß ein LGS gelöst und das Ergebnis den Variablen r,s,t zugewiesen werden.

```
> L:=solve({f(a)=p(a),f((a+b)/2)=p((a+b)/2),f(b)=p(b)},{r,s,t})
  ;
> assign(L);
```

$$L := \{s = \frac{-15}{2}, r = \frac{3}{2}, t = -1\}$$

Ermittlung des Polynoms:

```
> 'p(x)'=p(x);
```

$$\mathrm{p}(x) = \frac{3}{2}x^2 - \frac{15}{2}x - 1$$

b) Zeichnen der Graphen:

```
> plot({f(x),p(x)},x=-4..4,y=-15..15);
```



c) Vermutung: Die beiden von den Graphen eingeschlossenen Flächen besitzen gleichgroße Flächeninhalte!

```
> solve(f(x)=p(x),x);
```

$$2, \frac{1}{2}, -1$$

```
> Int('f(x)-p(x)',x=-1..0.5)=int(f(x)-p(x),x=-1..1/2);
  Int('p(x)-f(x)',x=0.5..2)=int(p(x)-f(x),x=1/2..2);
```

$$\int_{-1}^{.5} f(x) - p(x)\, dx = \frac{81}{64}$$

$$\int_{.5}^{2} p(x) - f(x)\, dx = \frac{81}{64}$$

Die Inhalte der beiden Differenzflächen sind gleich groß. Die Keplersche Fassregel liefert also auch für diese ganzrationale Funktion 3. Grades ein exaktes Ergebnis.

d) Zum allgemeinen Nachweis der Gleichheit der Differenzflächen:

```
> restart;
```

Eingabe der Funktion f mit unbestimmten Koeffizienten, des Näherungspolynoms und der Intervallgrenzen:

```
> f:=x->a3*x^3+a2*x^2+a1*x+a0;
> p:=x->r*x^2+s*x+t;
> a:=a;b:=b;
```

$$a := a$$

$$b := b$$

Wieder muss das Gleichungssystem gelöst werden und das Ergebnis den Variablen r,s,t zugewiesen werden.

```
> L:=solve({f(a)=p(a),f((a+b)/2)=p((a+b)/2),f(b)=p(b)},{r,s,t})
  ;
> assign(L);
```

$$L := \{t = a0 + \frac{1}{2} a3\, a\, b^2 + \frac{1}{2} a3\, a^2\, b,\ r = \frac{3}{2} a3\, a + \frac{3}{2} a3\, b + a2,$$
$$s = -\frac{1}{2} a3\, a^2 - 2\, a\, a3\, b + a1 - \frac{1}{2} a3\, b^2\}$$

Ermittlung des Polynoms:

```
> 'p(x)'=p(x);
```

$$p(x) = \left(\frac{3}{2} a3\, a + \frac{3}{2} a3\, b + a2\right) x^2 + \left(-\frac{1}{2} a3\, a^2 - 2\, a\, a3\, b + a1 - \frac{1}{2} a3\, b^2\right) x + a0 + \frac{1}{2} a3\, a\, b^2$$
$$+ \frac{1}{2} a3\, a^2\, b$$

Nun kann gezeigt werden:
Die beiden von den Graphen eingeschlossen Flächen besitzen den gleichen Flächeninhalt.

```
> solve(f(x)=p(x),x);
```

$$b,\ a,\ \frac{1}{2} a + \frac{1}{2} b$$

```
> Int('f(x)-p(x)',x=a..(a+b)/2)=factor(int(f(x)-p(x),x=a..(a+b)
  /2));
  Int('p(x)-f(x)',x=(a+b)/2..b)=factor(int(p(x)-f(x),x=(a+b)/2.
  .b));
```

$$\int_{a}^{1/2\,a+1/2\,b} f(x) - p(x)\,dx = \frac{1}{64}\,a3\,(a-b)^4$$

$$\int_{1/2\,a+1/2\,b}^{b} p(x) - f(x)\,dx = \frac{1}{64}\,a3\,(a-b)^4$$

Die Keplersche Fassregel liefert auch für ganzrationale Funktionen 3. Grades exakte Ergebnisse!

```
>
```

Aufgabe 5: Beweis der Vermutung aus Aufgabe 4

Der Beweis der Vermutung aus Aufgabe 4 kann auch durch Ermittlung der quadratischen Näherungsfunktion p für eine beliebige ganzrationale Funktion f vom Grad 3 geführt werden.
Ermittle diese und vergleiche das Ergebnis der Integration mit dem "exakten" Wert.
Gilt die Beobachtung auch für Funktionen vom Grad 4?

Lösungsvorschlag

```
> restart;
```

Eingabe der Keplerschen Fassregel:

```
> kepler:=(a,b)->(b-a)/6*(f(a)+4*f((a+b)/2)+f(b));
```

Eingabe der Funktion:

```
> f:=x->a3*x^3+a2*x^2+a1*x+a0;
```

Vergleich:

```
> 'kepler(a,b)'=factor(kepler(a,b));
> Int('f(x)',x=a..b)=factor(int(f(x),x=a..b));
```

$$\mathrm{kepler}(a,b) = -\frac{1}{12}(-b+a)\,(3\,a3\,a^3 + 4\,a2\,a^2 + 6\,a1\,a + 12\,a0 + 3\,a3\,a^2\,b + 3\,a3\,a\,b^2 + 3\,a3\,b^3 + 4\,a2\,a\,b + 4\,a2\,b^2 + 6\,a1\,b)$$

$$\int_a^b f(x)\,dx = -\frac{1}{12}(-b+a)\,(3\,a3\,a^3 + 4\,a2\,a^2 + 6\,a1\,a + 12\,a0 + 3\,a3\,a^2\,b + 3\,a3\,a\,b^2 + 3\,a3\,b^3 + 4\,a2\,a\,b + 4\,a2\,b^2 + 6\,a1\,b)$$

Beide Ergebnisse stimmen überein!

```
>
```

Der entsprechende Satz trifft für ganzrationale Funktionen 4. Grades nicht zu.
Widerlegung durch ein Gegenbeispiel:

```
> f:=x->2*x^4-x^3+2*x^2-4*x+2;
> 'kepler(-1,3)'=kepler(-1,3);
> Int('f(x)',x=-1..3)=int(f(x),x=-1..3);
```

$$\text{kepler}(-1, 3) = \frac{316}{3}$$

$$\int_{-1}^{3} f(x)\, dx = \frac{1324}{15}$$

```
>
```

Zusammenfassung der Ergebnisse

In diesem Arbeitsblatt haben wir ein Verfahren zur Berechnung von Näherungswerten für bestimmte Integrale hergeleitet. Dazu wurde die zu integrierende Funktion f auf dem Integrationsintervall [a,b] durch eine quadratische Funktion p approximiert. Anschließend wurde diese Näherungsfunktion über dem Intervall [a;b] integriert.
Wählt man die quadratische Näherungsfunktion p so, dass sie an den Grenzen und in der Mitte des Intervalls [a,b] mit der anzunähernden Funktion f übereinstimmt, so resultiert aus diesem Vorgehen die *Keplersche Fassregel*.
Obwohl nur eine quadratische Näherungsfunktion verwendet wird, ist die Keplersche Fassregel auch für ganz-rationale Funktionen vom Grad 3 exakt.

1. Bestimmung der quadratischen Näherungsfunktion p

Nach der Eingabe der Funktion f, des Näherungspolynoms p und der Intervallgrenzen ist das LGS zu lösen.
Die Ergebnisse werden den Variablen r,s,t zugewiesen. Die Funktion p kann nun berechnet werden.

```
> restart;
> f:=x->sin(x);
> p:=x->r*x^2+s*x+t;
> a:=0; b:=2;
> L:=solve({f(a)=p(a),f((a+b)/2)=p((a+b)/2),f(b)=p(b)},{r,s,t})
  ;
> assign(L);
> 'p(x)'=p(x);
>
```

2. Die Keplersche Fassregel

Nach Definition der Funktion f und der Keplerschen Fassregel können Näherungswerte für $\int_a^b f(x)\, dx$ berechnet werden durch $\int_a^b p(x)\, dx = \frac{(b-a)\left(f(a) + 4\, f\left(\frac{a+b}{2}\right) + f(b)\right)}{6}$.

```
> restart;
> f:=x->sin(x);
> kepler:=(a,b)->(b-a)/6*(f(a)+4*f((a+b)/2)+f(b));
> evalf(kepler(0,2));
```

>

Übungsaufgaben

1. Betrachte die Funktion f mit $f(x) = \sin(x^2)$ auf dem Intervall [0;1].
a) Bestimme die Funktionsgleichung der quadratischen Näherungsfunktion p.
b) Zeichne die Graphen von f und p in ein Koordinatensystem.
c) Berechne damit einen Näherungswert für $\int_0^1 \sin(x^2)\, dx$.
d) Wiederhole die Schritte a)-c) für das Intervall [0;1.5].

2. Zeige, dass die Keplersche Fassregel für die Funktion f mit
$f(x) = 4\,x^3 - 3\,x^2 + 2\,x - 5$ auf dem Intervall [-1;3] das exakte Ergebnis liefert.

3. Berechne $\int_0^{\pi} \sin(x)\, dx$ exakt und näherungsweise mit Hilfe der Keplerschen Fassregel.
Ermittle einen Näherungswert für π.

4. Leite die Keplersche Fassregel mit Hilfe eines CAS her.